

Staff Animation

(Animation de portée)

Cette application permet de faire des animations musicales (écrites), à des fins pédagogiques, pour les insérer dans des screencasts.

- [Animation](#)
- [Les notes](#)
- [Les accords](#)
- [Les portées](#)
- [Les gammes](#)
- [Les textes](#)
- [Les flèches](#)
- [Régler toutes les valeurs de l'animation](#)



Création d'une animation

Table des matières

- [Introduction](#)
- [Composer le code de l'animation](#)
 - [Commentaires dans le code](#)
 - [Faire une pause](#)
 - [Avancer sur la portée](#)
 - [Écrire un texte général](#)
 - [Resetter l'animation](#)
 - ["Nettoyer" l'animation \(tout effacer\)](#)
- [Vitesse de l'animation](#)
- [Activer/désactiver le Mode "Flash"](#)
- [Ré-initialiser toutes les valeurs de préférences](#)

Réinitialiser les préférences

Introduction

Une animation est composée de `pas` (step) qui vont être exécutés l'un après l'autre, produisant des choses aussi diverses que :

- L'apparition d'une portée ;
- L'écriture d'une note ou d'un accord sur la portée ;
- Le déplacement de notes ;
- L'écriture de textes explicatifs ;
- etc.

Chaque pas (chaque “step”) exécute une action. Ces pas se définissent dans la console située à gauche de l'écran (le bloc noir). Chaque pas se trouve sur une ligne distincte. Donc chaque ligne est un nouveau pas, une nouvelle étape de l'animation.

Bien noter que chaque ligne représente un pas, il est impossible de définir plusieurs actions sur la même ligne (en réalité c'est faux, mais c'est dangereux ;-)).

Ces pas peuvent :

- Construire une portée ;
- Construire une note (et l'afficher) ;
- Construire un accord ;
- Construire une gamme ;
- Écrire un texte associé à un objet (note, portée, barre, etc.) ;
- Déplacer un objet quelconque (des notes, des textes, etc.) ;
- Supprimer une note ;
- Mettre une note en exergue (entourée d'un cercle de couleur) ;
- etc.

Composer le code de l'animation

Écrire un commentaire

Les commentaires s'écrivent avec `#` en début de ligne.

Noter que les commentaires ne peuvent pas se trouver sur un pas proprement, même après le code. Il faut absolument qu'ils soient sur une ligne seule, qui sera passée.

Faire une pause dans l'animation.

On utilise la méthode `WAIT` pour faire une pause, avec en argument le nombre de secondes (qui peut être un flottant, pour des temps très courts) :

```
WAIT(<nombre de secondes>)
```

Par exemple, pour attendre 4 secondes :

```
maNote=NOTE(a3)
WAIT(4)           // 4 secondes avant de construire l
'accord
monAccord('c3 eb3 g3')
```

Note : C'est un "pas" comme les autres, donc il doit être mis sur une ligne seule comme toute étape.

Se déplacer sur la portée

La commande pour écrire à la suite des dernières notes sur la portée, on utilise la commande :

```
NEXT([<nombre pixels>])
```

Par défaut (sans argument), le déplacement sera de 40px (`Anim.default.hoffset`). Sinon, le déplacement sera de la valeur précisée.

Par exemple :

```
NEXT()
// => les notes suivantes s'éciront 40px plus à gauche

NEXT(100)
// => les notes suivantes s'éciront 100px plus à gauche
```

Écrire un texte général

Pour écrire un texte en rapport avec l'animation (en haut à gauche), on utilise la méthode `WRITE` :

`WRITE()`

Par exemple :

`WRITE("Ce qu'il faut remarquer à ce moment-là.")`

Noter que les objets tels que les notes ont également leur propre méthode textuelle, qui permet d'aligner le textes directement à ces notes, accords, etc.

Reset(-ter) l'animation

La commande `RESET` (sans parenthèses ni arguments) permet de repartir de rien en cours d'animation, c'est-à-dire :

- Effacer tous les objets SAUF les portées et les clés ;
- Remettre le "curseur" tout à gauche (endroit où seront donc affichés les nouveaux objets).
- Remettre toutes les valeurs par défaut.

Usage :

```
RESET
```

Voir aussi la commande `CLEAR` ci-dessous qui permet la même chose mais de façon plus ciblée.

Nettoyer l'animation

“Nettoyer l'animation” signifie supprimer tous les éléments affiché. On peut ou non demander que les portées restent en place, though.

Le pas à utiliser est :

```
CLEAR(<avec les portées>)
```

... `<avec les portées>` est FALSE par défaut, donc il faut ajouter `true` pour effacer aussi les portées :

```
CLEAR(true) // efface aussi les portées
```

Vitesse de l'animation

On peut régler la vitesse de l'animation de façon interactive avec le “slider” se trouvant dans le contrôleur (sous le bouton “Start”).

Pour régler la vitesse de façon très fine, cf. aussi [Réglage de la vitesse dans les préférences](#).

Mais on peut aussi définir dans le code cette vitesse à l'aide de la commande :

```
SPEED(<valeur>)
```

... avec `<valeur>` pouvant être un nombre de 0 à 20, `10` étant la vitesse normale, 0 étant la vitesse la plus lente et 20 la plus rapide.

Pour remettre la vitesse à la vitesse normale :

```
SPEED() // pas d'argument
```

Activer/désactiver le mode “Flash”

Le mode “Flash” permet de jouer très rapidement une partie de l'animation. C'est utile lorsqu'une partie est calée, mais qu'on ne veut pas la rejouer systématiquement au même rythme pour atteindre la portion en cours d'élaboration.

Dans ce cas, on met le code à passer entre :

```
MODE_FLASH
.....
... code ...
... à ...
... passer ...
.....
STOP_MODE_FLASH
```

Noter que le résultat ne sera pas du tout le même que si l'on commentait les lignes de code à passer, puisque l'animation ne se trouverait peut-être pas dans le même état.

Ré-initialiser toutes les valeurs de préférence

Pour remettre toutes les valeurs de décalages aux valeurs de départ, utiliser la commande (SANS PARENTHÈSES) :

```
RESET_PREFERENCES
```

Noter que ça ne ré-initialise que les décalages des éléments, tels que les textes de marque de l'harmonie ou des accords, etc. Pour une ré-initialisation complète, utiliser la commande [RESET](#) .

Les Notes

Table des matières

- [Désignation des notes](#)
 - [Les altérations](#)
- [Constantes notes](#)
- [Créer une note](#)
- [Déplacer une note](#)
- [Placer une note sur une portée précise](#)
- [Mettre une note en exergue/La sortir de l'exergue](#)
- [Entourer une note \(exergue plus forte\)](#)

- [Détruire d'une note](#)

Désignation des notes

Les notes doivent être désignées par :

```
<nom note><alteration><octave>
```

- On désigne le `nom des notes` par une seule lettre, anglaise, de "a" (la) à "g" (sol).
- L'altération est soit rien (note naturelle), soit un signe (cf. ci-dessous "b", "d", "x", ou "t").
- Vient ensuite l'octave, un nombre, négatif si nécessaire (*mais pour le moment, on va seulement jusqu'à l'octave 0, les autres ne sont pas gérés*).

Marque des altérations

La valeur `<alteration>` ci-dessus peut être :

- `b` pour bémol
- `t` (comme "ton") pour double bémol
- `d` pour dièse
- `x` pour double-dièse

Constantes notes

De l'octave 0 (qui n'existe pas en français) à l'octave 7 il existe des constantes pour chaque note avec l'altération bémol ("b") et dièse ("d").

Les deux formules suivantes sont donc possible :

```
no=NOTE ( ad5 )
no=NOTE ( ' ad5 ' )
```

Note: Attention à ne pas donner à une variable note le nom d'une de ces constantes. Par exemple, si on fait :

```
a5=NOTE ( a5 )
```

Cela générera une erreur de constante déjà définie.

On peut utiliser plutôt :

```
na5=NOTE(a5)
```

Créer une note

```
<variable name> = NOTE(<note>)
```

- `<variable>` peut avoir le nom qu'on veut, HORMIS un nom de constante, comme `a5`.
- La `<note>` peut être soit un string soit une constante (cf. [Désignation des notes](#))

On peut aussi, si l'on est certain que la note ne sera pas utilisée après (déplacée, supprimée, mise en exergue, etc.) ne pas l'affecter à une variable et composer immédiatement son aspect.

Par exemple, pour placer une note LA 4 avec une marque de cadence et un accord :

```
NEW_STAFF(SOL)  
NOTE(a4).cadence('V I').chord_mark('A Maj')
```

Déplacer une note

Pour déplacer une note, on utilise le pas :

```
<nom variable note>.moveTo(<nouvelle note>[,<params>])
```

Exemple :

```
maNote=NOTE(a4) // crée la note LA 4  
maNote.moveTo(g4) // descend la note vers SOL4
```

Ne mettre aucune espace dans ce code.

*Noter que la note de destination devra vraiment la nouvelle valeur de la note.
Si la note "a4" se déplace vers "a3", cette note deviendra "a3" dans ses données.*

Placer une note sur une portée précise

Si on veut placer une note sur une portée hors de la portée active, on indique l'indice de cette portée avant la note, puis ":" :

```
<indice portée>:<note>
```

Par exemple :

```
note_autre_staff=NOTE( '2:a4' )  
OU  
note_autre_staff=NOTE( '2:'+a4 )
```

... placera un LA4 sur la deuxième portée, même si elle n'est pas active.

Noter que cela ne rend pas la portée active.

Mettre une note en exergue (ou la retirer de l'exergue)

Utiliser la méthode `exergue()` (pour mettre en exergue, en couleur) et `unexergue()` (pour la sortir de l'exergue).

Exemple :

```
maNote=NOTE( 'cx5' )  
maNote.exergue()    # => note en couleur  
WAIT(4)  
maNote.unexergue()  # => etat normal
```

Cf. aussi [Entourer une note](#) ci-dessous.

Entourer une note

Pour mettre en exergue une note de façon plus forte que la méthode [exergue](#), on peut utiliser la méthode `surround()` qui entoure la note d'un cercle de

couleur.

Syntaxe :

```
<note>.surround([<parameters>])
```

... où `<note>` est une instance de Note, et `<parameters>` sont les paramètres optionnels envoyés.

Par exemple :

```
maNote=NOTE(c5)
maNote.surround() # => entoure d'un cercle rouge

oNote=NOTE(a4)
oNote.surround({color:blue, rectangle:true})
# => entoure d'un rectangle bleu
```

Les paramètres peuvent être :

color:{String|Constante}

La couleur, parmi 'blue', 'red', 'orange', 'green', 'black' (les valeurs peuvent se définir avec ou sans guillemets).

rectangle:{Boolean}

Si true, un rectangle au lieu d'un rond.

margin:{Number}

Joue sur le diamètre du cercle/rectangle pour laisser plus ou moins de place. Utile lorsque plusieurs notes assez proches sont entourées

Retirer le cercle

Deux commandes pour retirer le cercle de la note :

```
maNote.circle.remove()
```

Ou :

```
maNote.unsurround()
```

Détruire une note

Pour détruire la note (la supprimer de l'affichage, utiliser :

```
<nom variable note>.remove()
```

Les Accords

Table des matières

- [Création d'un accord](#)
- [Référence aux notes de l'accord](#)
- [Écrire le nom de l'accord](#)
- [Écrire l'harmonie](#)
- [Écrire une cadence](#)
- [Écrire une modulation](#)
- [Destruction d'un accord](#)

Création d'un accord

On crée un accord avec :

```
monAccord=CHORD( '<note1> <note2>... <noteN>' )
```

... où chaque note doit correspondre à la définition normale.

Par exemple :

```
accDom=CHORD( 'c3 eb3 g3' )
```

Création d'un accord sur plusieurs portées

On peut poser l'accord sur différentes portées en ajoutant

`<indice portée>:` devant la note (*rappel : l'indice portée est "1-start",*

donc "1" pour la première portée en comptant depuis le haut).

Noter qu'il est inutile d'indiquer l'indice portée de la portée active.

Par exemple (en imaginant que la portée 1 est la portée active) :

```
acc=CHORD('2:c3 2:g3 e4 g4')
```

... placera "c3" et "g3" sur la 2^e portée (certainement la clé de fa) et les notes "e4" et "g4" sur la 1^{ère} portée qui est la portée active.

Référence aux notes de l'accord

Comme tous les types d'objets possédant plusieurs notes (Accords, Gammes, Motifs), on fait appel aux différentes notes à l'aide de la méthode `note` appelée sur l'objet :

```
<nom accord>.note(<indice note>)
```

... où `<indice note>` est l'indice 1-start (1 = première note fournie).

Par exemple :

```
accDom=CHORD('c3 eb3 g3')
accDom.note(1).moveTo('c4')
// Prends la première note (c3) et la déplace en c4.
```

Comme pour tout "groupe de notes", si l'on veut affecter une note de l'accord à une variable (pour l'utiliser plus facilement ensuite), on ne peut pas procéder ainsi :

```
maNote=monAccord.note(2)
maNote.write("Une seconde !") # => # ERREUR #
```

Pour ce faire, il faut impérativement utiliser :

```
maNote=Anim.Objects.monAccord.note(2)
maNote.write("Une seconde !")
```

Donc ajouter `Anim.Objects` devant `monAccord.note(2)` ,
`Anim.Objects` étant la propriété qui contient tous les objets de
l'animation.

Écrire le nom de l'accord

Une marque d'accord est un texte. Cf. [Écrire un accord](#).

Écrire l'harmonie

Une marque d'harmonie est un texte. Cf. [Écrire l'harmonie](#).

Écrire une cadence

Une cadence est un texte. Cf. [Écrire une cadence](#).

Écrire une modulation

Une marque de modulation est un texte. Cf. [Écrire une modulation](#).

Destruction d'un accord

Pour détruire l'accord, utiliser :

```
<nom variable accord>.remove()
```

Les portées

Table des matières

- [Créer une portée](#)
- [Activer une portée](#)
- [Récupérer une portée](#)
- [Supprimer les lignes supplémentaires](#)

Créer une portée

Pour créer une portée, utiliser le pas :

```
NEW_STAFF(<cle>[ , <params>])
```

... où `<cle>` peut être `SOL` , `FA` , `UT3` , `UT4` .

... et `<params>` peut contenir :

```
NEW_STAFF(<cle>, {  
    offset : decalage supplementaire par rapport à précédente  
             (pixels)  
})
```

Noter que par défaut, il y aura toujours un décalage entre deux portées créées, ce `offset` ne fait qu'aggrandir l'espacement (s'il est positif) ou le rétrécir (s'il est négatif)

Par exemple :

```
NEW_STAFF ( SOL )
```

... qui affichera une portée en clé de sol en dessous de la dernière portée.

Noter que cette portée deviendra la portée active, c'est-à-dire celle où seront placées les objets définis par la suite.

Activer une portée

Activer une portée signifie que tous les pas suivants la viseront. Par exemple, les notes se déposent toujours sur la portée active.

```
ACTIVE_STAFF(<indice de la portee>)
```

... où `<indice de la portee>` est son rang dans l'affichage, en partant de 1 et du haut. Donc la portée la plus en haut s'active par :

```
ACTIVE_STAFF ( 1 )
```

Récupérer une portée

On peut récupérer une portée (pour la “travailler” dessus) à l'aide de la commande :

```
STAFF(<indice 1-start>)
```

Par exemple, pour supprimer la deuxième portée construite :

```
STAFF(2).remove()
```

Noter que pour mettre la portée dans une variable, il faut impérativement utiliser :

```
maPortee=Anim.Objects.STAFF(x)
```

Supprimer des lignes supplémentaires

Pour le moment, la suppression de lignes supplémentaires n'est pas automatique, afin de laisser toute liberté à la programmation de l'animation.

Utiliser la commande :

```
STAFF(<indice 1-start>).remove_suplines({
    top      : <indice ou liste d'indice a supprimer au-dessus
>,
    bottom   : <indice ou liste d'indices a supprimer en dessous>,
    xoffset  : <decalage horizontal>
})
```

- N'utiliser que `top` et `bottom` au besoin ;
- Les indices des lignes se comptent À PARTIR DE LA PORTÉE (donc en montant pour `top` et en descendant pour `bottom`) ;
- `xoffset` n'est à préciser que si les lignes à supprimer ne se trouvent pas sur le décalage horizontal courant (suppression arrière, rare).

Autre commande pour supprimer des lignes supplémentaires

On détruit ces lignes à l'aide de la commande :

```
REMOVE_SUPLINE(<parameters>)
```

Une ligne supplémentaire est caractérisée par :

- La portée qui la porte ;
- Son indice à partir de la portée ;
- Sa position supérieure ou inférieure ;
- Son décalage à gauche ("frame" de l'animation).

Cela détermine les paramètres de `<parameters>` .

```
{
  staff: indice 1-start de la portée (depuis le haut),
  bottom: liste d'indices ou indice de la ligne à supprimer
en bas,
  top: liste d'indices ou indice de la ligne à supprimer en
haut,
  xoffset: décalage gauche (frame)
}
```

Toutes les valeurs à part `bottom` xou `top` sont optionnelles :

Si `staff` n'est pas précisé, on prendra la portée active.

Si `xoffset` n'est pas précisé, on prendra le décalage courant (ce qui représente le cas le plus fréquent, entendu qu'on va rarement supprimer une ligne supplémentaire "en arrière").

Note : lors d'un déplacement, une suppression ou tout autre effet qui doit rendre obsolète la ligne supplémentaire, il est préférable de déclencher la suppression des lignes supplémentaires AVANT la commande sur la note. Par exemple, pour un déplacement :

```
note=NOTE(c4) // ajoute une ligne supplément en bas
WAIT(2)
REMOVE_SUPLINE({bottom:1})
note.moveTo(c5)
```

Précision des indices

Les indices peuvent être une simple valeur numérique :

```
bottom:1 / top:1
```

... ou une liste d'indices

```
top:[1,2] / bottom:[1,2]
```

Noter que ces indices sont "1-start" et se comptent toujours À PARTIR DE LA portée, donc en descendant pour `bottom` et en montant pour `top`.

Noter aussi que `bottom` et `top` sont complètement indépendants, pour `bottom` on ne tient compte QUE des lignes supplémentaires inférieures et pour `top` on ne tient compte QUE des lignes supplémentaires supérieures.

////////////////////////////////////

Les gammes

Table des matières

- [Introduction](#)
- [Paramètres de définition des gammes](#)
- [Utilisation des notes de la gammes](#)

Introduction

On peut produire en un seul pas une gamme à l'aide de la commande :

```
<var>=SCALE(<tonalité>[, <paramters>])
```

- `<var>` est un nom de variable quelconque
- `<tonalité>` est la tonalité exprimée par une seule lettre (anglaise) de "a" (la) à "g" (sol). On peut ajouter toutes les altérations voulues (cf. [Les altérations](#)). Noter que par défaut, suivant la portée active, l'animation affiche ses notes à la hauteur où elles produiront le moins de lignes supplémentaires.
- `<parameters>` est une liste de paramètres optionnels. Cf. ci-dessous.

Paramètres de définition des gammes

Ils constituent le second argument de la commande `SCALE`, après la note de la tonalité.

C'est un objet de propriétés :

```
maGamme=SCALE( ' a ' , {  
    octave : 2 ,  
    for     : 5 ,  
    etc.  
} )
```

Liste des propriétés

type : {String}

Le type de gamme, parmi : 'MAJ' : Gamme majeure (défaut), 'min_h' : Mineure harmonique, 'min_ma' : MINEure Mélodique Ascendante, 'min_md' : MINEure Mélodique Descendante.

octave : {Number}

L'octave à laquelle il faut afficher la gamme. Par défaut, celui qui produira le moins de ligne supplémentaires, donc celui dont le plus grand nombre de notes se trouve *dans* la portée.

staff : {Number}

L'indice de la portée sur laquelle il faut écrire la gamme. Par défaut, la portée active.

offset : {Number}

Le décalage horizontal pour commencer la gamme. Par défaut le décalage courant (la position du "pointeur").

asc : {Boolean}

Si TRUE (défaut), la gamme sera ascendante, sinon, elle descendra.

for : {Number}

Le nombre de notes de la gamme à afficher. Par défaut, 8 pour pouvoir

////////////////////////////////////

Les Textes

Table des matières

- [Introduction aux textes](#)
- [Créer un texte](#)
 - [Types spéciaux de texte \(accord, harmonie, etc.\)](#)
 - [Créer un texte pour l'animation](#)
 - [Créer un texte pour un objet](#)
- [Supprimer un texte](#)
 - [Supprimer le texte d'un objet](#)
- [Définir les positions des textes](#)

Introduction

Les textes peuvent exister pour l'animation en général (ils sont alors écrits en haut à gauche et chaque nouveau texte remplace l'ancien), mais ils peuvent être associés aussi à tout objet de l'animation, note, portée, mesure, barre, etc.

Créer un texte

Types spéciaux de texte

Par défaut, un texte est "normal", il s'écrit tel qu'il est défini.

Mais il existe des types qui peuvent être définis grâce à la propriété `type` envoyée en paramètres :

chord

Le type ``chord`` permet d'écrire un accord au-dessus de l'élément porteur du texte. Il est stylisé en conséquence. Cf. [Écrire un accord](#).

harmony

Écrit le texte sous la portée, sous forme d'une marque d'harmonie.
Si le texte se finit par un certain nombre de "*" ou de "•", ils sont considérés comme des renversements de l'accord et traités visuellement comme tels. Cf. [Écrire l'harmonie](#).

cadence

Écrit le texte sous la portée, à la position courante, sous forme de marque cadentielle (donc avec des traits "___|" pour marquer la fin de la partie). Cf. [Écrire une cadence](#).

modulation

Écrit le texte pour une modulation, au-dessus de la portée et de travers.
Cf. [Écrire une modulation](#).

Écrire l'harmonie

Pour indiquer une harmonie, par exemple pour indiquer que l'accord est un premier degré sous la forme de son deuxième renversement :

```
monAccord=CHORD('g4 c5 e5')  
monAccord.write("I**",{type:harmony})
```

Il existe aussi le raccourci :

```
monAccord.harmony("I**")
```

Pour le positionnement de la marque, cf. [Position des textes d'harmonie et de cadence](#).

Noter que pour allonger la barre inférieur de la marque d'harmonie, on peut ajouter le paramètre `width` qui sera le nombre de pixels désiré. Par exemple :

```
monAccord.write("I**", {type:harmony, width:100})  
OU  
monAccord.harmony("I**", {width:100})
```

Écrire une cadence

Pour indiquer une **CADENCE** :

```
monAccord.write("I", {type:cadence})  
OU  
monAccord.cadence("I")
```

Pour le positionnement de la marque, cf. [Position des textes d'harmonie et de cadence](#).

Écrire un accord

Pour indiquer le **NOM DE L'ACCORD** au-dessus des notes (ou d'un autre objet) :

```
monAccord.write("C", {type:chord})
```

... ou le raccourci `chord_mark` :

```
monAccord.chord_mark("C")
```

Pour le positionnement de la marque de l'accord cf. [Réglage de la position de l'accord](#).

Écrire une modulation

Pour écrire une modulation (au-dessus de la portée, nom de l'accord de travers), utiliser :

```
<objet>.modulation(<nom>[, <parameters>])
```

Par exemple :

```
monAccord = CHORD('c4 d4 fd4 a4')  
monAccord.modulation('Sol')
```

Ou :

```
monAccord.write('Sol', {type:modulation})
```

Texte sous la barre

Le ton de la modulation est indiqué au-dessus de la barre transversal, mais un autre texte peut être écrit SOUS la barre. Pour ce faire, il suffit de le mettre après l'accord de la tonalité, en le séparant d'une espace.

```
monAccord.modulation('Sol Dom de Dom')
```

... produira quelque chose comme (mais incliné) :

```
| SOL
|-----
| Dom de Dom
|
|
```

Pour le positionnement de la marque de modulation cf. [Réglage de la position de la marque de modulation](#).

Créer un texte pour l'animation

Utiliser la commande :

```
WRITE("<le texte>")
```

Créer un texte pour un objet

Pour associer un texte à un objet, il faut bien sûr créer l'objet puis ensuite appeler sa méthode `write` (écrire) :

```
maNote=NOTE(a4)
maNote.write("C'est un LA 4")
```

Supprimer un texte

Supprimer un texte d'objet

Pour supprimer le texte de l'objet, c'est-à-dire de le faire disparaître de l'affichage, utiliser la méthode `hide` (cacher) ou `remove` du texte de

l'objet :

```
<note>.texte[<type>].hide() / remove()
```

Où `<type>` est le type de texte :

- 'regular' pour un texte normal (sans type)
- 'chord' pour un accord
- 'cadence' pour une cadence
- 'harmony' pour une harmonie
- 'modulation' pour une modulation.

Par exemple :

```
maNote=NOTE(a4)
# Écrire le texte
maNote.write("C'est un LA 4")
# Attendre 2 secondes
WAIT(2)
# Supprimer le texte
maNote.texte['regular'].hide()
```

Noter que cette méthode supprime l'affichage du texte, mais l'objet `texte` existe toujours pour l'objet et on peut le ré-utiliser plus tard, par exemple avec :

```
maNote.texte['regular'].show()
```

... qui ré-affichera ce texte.

En revanche, si on utilise :

```
maNote.texte['regular'].remove()
```

... alors l'objet sera vraiment détruit.

//

Les flèches

Table des matières

- [Introduction aux flèches](#)
- [Associer une flèche à un objet](#)
- [Définition de la flèche \(paramètres\)](#)
- [Méthodes d'animation des flèches](#)
- [Angle des flèches](#)
- [Détruire la flèche](#)
- [Flèches indépendantes](#)

Introduction

On peut créer des flèches indépendantes (cf. [Flèches indépendantes](#)) mais le plus judicieux est de les associer à des objets, à commencer par des notes.

Associer une flèche à un objet

Pour associer une flèche à un objet quelconque (p.e. une note) on utilise la méthode (de l'objet) :

```
arrow(<parameters>)
```

Par exemple, pour faire partir une flèche d'une note :

```
maNote=NOTE(a4)  
maNote.arrow({color:red, angle:90})
```

... ce qui produira une flèche rouge, partant de la note avec un angle de 90 degré (cf. ci-dessous [les angles](#)).

Définition de la flèche

Lors de la création de la flèche avec la méthode `arrow` (ou `ARROW` pour une [flèche indépendante](#)) on peut envoyer ces paramètres optionnels à la méthode :

```
maNote.arrow({  
  width:  {Number} longueur fleche en pixels  
  angle:  {Number} Angle en degres  
  color:  {String} La couleur (constante ou string)  
  top:    {Number} Placement vertical de la fleche (pixels)  
  left:   {Number} Placement horizontal de la fleche (pixels)  
  height: {Number} Hauteur de la fleche  
})
```

Les valeurs **top** et **left** sont calculées automatiquement pour que la flèche soit placée correctement à droite de l'objet.

L'**angle** est de 0 degré par défaut, c'est-à-dire que la flèche sera horizontale et pointera à droite (pour une autre valeur cf. [Angle des flèches](#)).

La **color** (couleur) est noire par défaut. On peut utiliser toutes les constantes de couleur pré-définies, c'est-à-dire (les valeurs françaises sont des valeurs valides aussi) :

```
black    noir  
red      rouge  
blue     bleu  
green    vert  
orange
```

Méthodes d'animation des flèches

- [Faire tourner la flèche](#)
- [Changer la taille de la flèche](#)
- [Déplacer la flèche](#)

Faire tourner la flèche

```
<fleche>.rotate(<angle>)
```

Par exemple, pour une flèche associée à une note :

```
maNote=NOTE(c4)
maNote.arrow()
maNote.arrow.rotate(45)
```

Changer la taille (longueur) de la flèche

```
<fleche>.size(<longueur de la fleche>)
```

Par exemple :

```
maNote=NOTE(b4)
maNote.arrow({color:bleu})
maNote.arrow.size(100)
```

Produira une animation qui fera s'allonger la flèche de sa longueur actuelle à la longueur `100px` .

Noter que cette méthode **crée réellement une animation** c'est-à-dire fait varier sous nos yeux la taille de la flèche. Si on veut définir la taille de la flèche au départ, utiliser plutôt le paramètre `width` dans les paramètres envoyés à la création de la flèche (cf. [définition de la flèche](#)).

Déplacer la flèche

Une flèche se déplace à l'aide de la méthode :

```
<fleche>.move(<parameters>)
```

... où les paramètres peuvent être :

```
move({
  x : {Number} Déplacement horizontal (en pixels)
  y : {Number} Déplacement vertical (en pixels)
})
```

Par exemple, si une flèche associée à une note doit se déplacer en descendant et en se déplaçant vers la droite :

```
maNote=NOTE(g4)
maNote.arrow()
WAIT(2)
maNote.arrow.move({x:50, y:50}) <--
```

Une valeur positives produira toujours un déplacement vers la droite (->) pour `x` et un déplacement vers le bas pour `y`, un valeur négative produira un déplacement vers la gauche (<-) pour `x` et un déplacement vers le haut pour `y`.

Angle des flèches

Repère pour la définition de l'angle d'une flèche :

```
angle = 0    => flèche horizontale pointant à droite
angle = 90   => flèche verticale pointant en bas
angle = -90  => flèche verticale pointant en haut
angle = 180  => flèche horizontale pointant à gauche
```

Noter que pour les valeurs entre 90 et -90 (donc pointant vers la gauche), il faut modifier le `left` de la flèche pour qu'elle ne traverse pas la note.

Détruire la flèche

Pour détruire la flèche (la retirer de l'affichage), utiliser sa méthode `remove`.

Par exemple :

```
maNote=NOTE(c3)
maNote.arrow()
WAIT(2)
maNote.arrow.remove() <--
```

Flèches indépendantes

[TODO]

////////////////////////////////////

Régler les valeurs par défaut

On peut régler à tout moment (et en particulier au début de l'animation) toutes les valeurs par défaut, particulièrement tout ce qui concerne les positionnements.

Table des matières

- [Réglage de la vitesse](#)
- [Astuce pour le réglage des positions](#)
- [Réglage de la position des portées](#)
- [Réglage de la position horizontale initiale \(curseur\)](#)
- [Réglage de l'avancée à chaque NEXT \(curseur\)](#)
- [Position des textes d'harmonie et de cadence](#)
- [Position des marques d'accord](#)
- [Position de la marque de modulation](#)
- [Taille des notes](#)
- [Ré-initialiser toutes les valeurs par défaut](#)

Réglage de la vitesse

```
DEFAULT({speed: <coefficient>})
```

Ou :

```
DEFAULT('speed', <coefficient>)
```

... où `<coefficient>` est un nombre différent de zéro, qui accélère l'animation s'il est < 1 (p.e. `0.5`) et qui la ralentit s'il est supérieur à 1 (p.e. `5`)

Pour remettre les valeurs par défaut

```
DEFAULT('speed')
```

Astuce pour le réglage des positions

Ci-dessous, on peut découvrir le fonctionnement de la commande `DEFAULT` pour redéfinir des valeurs de l'animation, et notamment des positionnements d'élément.

Pour pouvoir trouver la valeur-pixel à affecter, le mieux est de procéder ainsi :

1. Trouver dans ce manuel le paramètre à modifier pour obtenir le positionnement recherché.
2. Une fois ce paramètre déterminé, regarder sa valeur actuelle dans le bloc des infos de l'animation (qui doit se trouver en bas à droite de la page). Au bout d'une ligne commençant par ce paramètre, vous pouvez trouver sa valeur.
3. Ensuite, déplacer la règle de mesure (en bas à gauche de l'animation — l'animation, pas la page) à l'endroit du changement pour estimer les nouvelles valeurs à donner (50, 100 et 150 sont en pixels).
4. Régler le paramètre avec la valeur estimée et essayer.

Réglage de la position des portées

Pour régler la position verticale de la première portée :

```
DEFAULT('staff_top', <nombre de pixels>)
```

Ou :

```
DEFAULT({staff_top: <nombre de pixels>})
```

Pour remettre la valeur par défaut :

```
DEFAULT('staff_top')
```

Noter que si cette valeur est modifiée en cours d'animation (à la volée), cela affectera la position des portées suivantes.

Pour régler l'espace vertical entre les portées :

```
DEFAULT('staff_offset', <nombre de pixels>)
```

Ou :

```
DEFAULT({staff_offset: <nombre de pixels>})
```

Pour remettre la valeur par défaut :

```
DEFAULT('staff_offset')
```

Réglage de la position horizontale initiale

Cette position correspond à la position du “curseur”, c'est-à-dire l'endroit où seront marqués les premiers éléments sur la portée (hors clé, armure et métrique).

```
DEFAULT('x_start', <nombre pixels>)
```

Ou :

```
DEFAULT({x_start: <nombre pixels>})
```

Pour remettre la valeur par défaut :

```
DEFAULT('x_start')
```

Pour déterminer facilement la valeur, cf. [Astuces pour le réglage des positions](#).

Réglage de la position **next**

La position **next** détermine de combien de pixels le “curseur” se déplacera à droite lorsque la commande **NEXT** sera utilisée (déterminant la position où seront créés les prochains objets)

Pour la régler de façon absolue

```
DEFAULT({next: <nombre pixels>})
```

Ou :

```
DEFAULT('next', <nombre pixels>)
```

Pour la régler de façon relative

(par rapport au `next` courant)

```
DEFAULT({offset_next: <nombre de pixels>})
```

Ou :

```
DEFAULT('offset_next', <nombre pixels>)
```

Pour remettre les valeurs par défaut

```
DEFAULT('next')  
DEFAULT('offset_next')
```

Pour déterminer facilement la valeur, cf. [Astuces pour le réglage des positions](#).

Réglage de la position des textes d'harmonie et de cadence

(Pour les marques d'harmonie et de cadence, cf. [Types spéciaux de texte](#))

Pour la régler de façon absolue

```
DEFAULT({harmony: <nombre de pixels>})
```

Ou :

```
DEFAULT('harmony', <nombre pixels>)
```

Pour la régler de façon relative (par rapport à décalage courant)

```
DEFAULT({offset_harmony: <nombre de pixels>})
```

Ou :

```
DEFAULT('offset_harmony', <nombre pixels>)
```

On peut utiliser aussi la commande spéciale :


```
OFFSET_HARMONY(<nombre de pixels>)
```

Pour remettre les valeurs par défaut

```
DEFAULT('harmony')  
DEFAULT('offset_harmony')
```

Pour déterminer facilement la valeur, cf. [Astuces pour le réglage des positions](#).

Réglage de la position des marques d'accord

(Pour les marques d'accords, cf. [Types spéciaux de texte](#))

Pour régler la position de façon absolue

```
DEFAULT({chord_mark: <nombre de pixels>})
```

Ou :

```
DEFAULT('chord_mark', <nombre pixels>)
```

Pour régler la position de façon relative (par rapport à décalage courant)

```
DEFAULT({offset_chord_mark: <nombre de pixels>})
```

Ou :

```
DEFAULT('offset_chord_mark', <nombre pixels>)
```

On peut utiliser aussi la commande spéciale :

```
OFFSET_CHORD_MARK(<nombre de pixels>)
```

Pour remettre les valeurs par défaut

```
DEFAULT('chord_mark')  
DEFAULT('offset_chord_mark')
```

Pour déterminer facilement la valeur, cf. [Astuces pour le réglage des positions](#).

Position de la marque de modulation

Pour placer la marque de modulation, on peut utiliser pour la position verticale et la position horizontale une valeur absolue ou une valeur relative (relative aux positions actuelles).

Régler la position horizontale absolue

```
DEFAULT('modulation_x', <nombre de pixels>)
```

Ce `<nombre de pixels>` sera retiré au `left`, donc si le nombre est positif, la marque recule vers la gauche.

Pour revenir à la position par défaut :

```
DEFAULT('modulation_x')
```

Pour déterminer facilement la valeur, cf. [Astuces pour le réglage des positions](#).

Régler la position horizontale relative

Pour modifier la position relativement à la position courante :

```
DEFAULT('offset_modulation_x', <nombre de pixels de decalage>)
```

Régler la position verticale absolue

```
DEFAULT('modulation_y', <pixels par rapport au haut de la portée>)
```

Régler la position verticale relative

```
DEFAULT('offset_modulation_y', <décalage par rapport à la position courante>)
```

Un nombre positif baissera la marque, un nombre négative la remontera.

Réglage de la taille des notes

Note : C'est une donnée qu'il vaut mieux ne pas modifier puisqu'elle est calculée automatiquement par rapport à la taille de la portée.

```
DEFAULT('note_size', <hauteur en pixel (flottant)>)
```

Ou :

```
DEFAULT({note_size: <hauteur en pixel>})
```

Pour remettre la valeur par défaut :

```
DEFAULT('note_size')
```

Ré-initialiser toutes les valeurs par défaut

Pour remettre toutes les valeurs aux valeurs par défaut, utiliser la commande (SANS PARENTHÈSES) :

```
RESET_PREFERENCES
```

-